
Flint Feature Walkthrough
    

Mix.install(
  [
    {:flint, "~> 0.5"},
    {:poison, "~> 6.0"}
  ],
  consolidate_protocols: false
)

  
    
  
  Introduction to Flint


Flint is a library that aims to make Ecto embedded_schema's more declarative, flexible, and expressive.
One of the core tenets of Flint is to be a drop-in replacement for Ecto, meaning that for all of your Ecto embedded schemas you can just switch from use Ecto.Schema to use Flint.Schema, and then you have to opt into all of the extra features of Flint.
Flint aims to empower you to colocate as much information as possible into your embedded_schema definitions, but Flint core is very unopinionated about how you use that data. Flint provides several build in extensions that are opinionated, but fundamentally Flint just exposes a way for you to store and retrieve additional data in an Ecto embedded_schema.

  
    
  
  Core Features


The core features of Flint are those not packaged as Flint extensions.
All of these features are available even when specifying extensions: [] when using Flint.Schema
	Added field!, embeds_one!, and embeds_many! macros that tag those fields as required and exposes them through the __schema__(:required) reflection function.
	Added support for do block in field and field! to add validation_condition -> error_message pairs to the field.
	Generated functions:	changeset/3
	new/2
	new!/2


	Custom implementation of embedded_schema that uses the above macros from Flint.Schema instead of Ecto.Schema.
	The ability to define aliases for field options
	Application-wide default options using configurations

Let's explore each of these in more detail

  
    
  
  Using Generated Functions



  
    
  
  changeset, new, and new!


Flint provides generated and overridable default implementations of changeset, new, and new! functions.
The new and new! functions use the changeset function.
The generated changeset function automatically accounts for the required fields, and now you can use changeset as you would any other changeset.
Many of the built-in extensions (which we will discuss in-depth later) build upon the core changeset function to build up a more comprehensive pipeline.
You can also use the generated new and new! functions. new will create a new struct from the passed params and will apply the changes regardless of validation, as opposed to new!, which will raise on validation errors, but otherwise will apply any valid changes.
Let's take a look at a more practical example. In this example, we're:
	Using both normal and ! variants of field declarations
	Using the shorthand notation where we pass the schema as an option to the use Flint call
	Using both external and inline embeds fields.
	Using an Ecto.Enum field type to map values between embedded and dumped representations.

defmodule Book do
  use Flint.Schema,
    schema: [
      field!(:title, :string),
      embeds_one! :author, Author_d do
        field!(:first_name)
        field!(:last_name)
        field(:bio, :string)
      end,
      embeds_many(:coauthors, Author_c),
      field(:genre, Ecto.Enum, values: [biography: 0, science_fiction: 1, fantasy: 2, mystery: 3])
    ]
end
{:module, Book, <<70, 79, 82, 49, 0, 0, 32, ...>>, :ok}
Now when we call Book.new it will create a new Book struct regardless of validation errors.
Note that by embeds_many(!) fields will default to an empty list ([]) at all times, whereas embeds_one! defaults to nil as it marks the field as :required, whereas embeds_one defaults to the empty struct (of its embedding). You can control this behavior for embeds_one using the defaults_to_struct boolean option.
Book.new()
%Book{title: nil, author: nil, coauthors: [], genre: nil}
Book.__schema__(:required)
[:title, :author]
Book.new!(%{title: "The old man and the sea"})
The generated changeset functions will also enforce :required validations for embedded fields, so if any required field of the :author embedded field is not present, then Book will fail validation.
Book.new!(%{
  title: "Harry Potter",
  author: %{first_name: "J.K."},
  genre: :fantasy
})
book =
  Book.new!(%{
    title: "Harry Potter",
    author: %{first_name: "J.K.", last_name: "Rowling"},
    genre: :fantasy
  })
%Book{
  title: "Harry Potter",
  author: %Book.Author_d{first_name: "J.K.", last_name: "Rowling", bio: nil},
  coauthors: [],
  genre: :fantasy
}
defmodule Book_b do
  use Flint.Schema,
    schema: [
      field!(:title, :string),
      embeds_one :author, Author_c do
        field!(:first_name)
        field!(:last_name)
        field(:bio, :string)
      end,
      embeds_many(:coauthors, Author_d),
      field(:genre, Ecto.Enum, values: [biography: 0, science_fiction: 1, fantasy: 2, mystery: 3])
    ]
end
{:module, Book_b, <<70, 79, 82, 49, 0, 0, 33, ...>>, :ok}
Note the quirk that if the embedded field is not marked as required, but one of its subfields is, then if the embedded struct is partially initialized it will fail if the required subfield is missing
Book_b.new!(%{
  title: "Harry Potter",
  author: %{first_name: "J.K."},
  genre: :fantasy
})
Whereas if it is not passed altogether then it will pass validation.
Book_b.new!(%{
  title: "Harry Potter",
  genre: :fantasy
})
%Book_b{
  title: "Harry Potter",
  author: %Book_b.Author_c{first_name: nil, last_name: nil, bio: nil},
  coauthors: [],
  genre: :fantasy
}

  
    
  
  Required Fields


defmodule Person do
  use Flint.Schema, extensions: []

  @primary_key false
  embedded_schema do
    field! :first_name, :string
    field :last_name, :string

    embeds_many! :parents, Parent, primary_key: false do
      field! :relationship, Ecto.Enum, values: [:mother, :father]
      field! :first_name, :string
      field :last_name, :string
    end
  end
end
{:module, Person, <<70, 79, 82, 49, 0, 0, 21, ...>>, :ok}
Now we can use the generated function to create a new Person struct
mark_twain = %{
  first_name: "Mark",
  last_name: "Twain",
  parents: [
    %{
      first_name: "John",
      last_name: "Clemens",
      relationship: :father
    },
    %{
      first_name: "Jane",
      last_name: "Clemens",
      relationship: :mother
    }
  ]
}

Person.changeset(%Person{}, mark_twain)
#Ecto.Changeset<
  action: nil,
  changes: %{
    parents: [
      #Ecto.Changeset<
        action: :insert,
        changes: %{first_name: "John", last_name: "Clemens", relationship: :father},
        errors: [],
        data: #Person.Parent<>,
        valid?: true,
        ...
      >,
      #Ecto.Changeset<
        action: :insert,
        changes: %{first_name: "Jane", last_name: "Clemens", relationship: :mother},
        errors: [],
        data: #Person.Parent<>,
        valid?: true,
        ...
      >
    ],
    first_name: "Mark",
    last_name: "Twain"
  },
  errors: [],
  data: #Person<>,
  valid?: true,
  ...
>
This generates an Ecto.Changeset. Let's remove one of the fields that had been marked as required.
mark_twain_bad = Map.drop(mark_twain, [:first_name])
Person.changeset(%Person{}, mark_twain_bad)
#Ecto.Changeset<
  action: nil,
  changes: %{
    parents: [
      #Ecto.Changeset<
        action: :insert,
        changes: %{first_name: "John", last_name: "Clemens", relationship: :father},
        errors: [],
        data: #Person.Parent<>,
        valid?: true,
        ...
      >,
      #Ecto.Changeset<
        action: :insert,
        changes: %{first_name: "Jane", last_name: "Clemens", relationship: :mother},
        errors: [],
        data: #Person.Parent<>,
        valid?: true,
        ...
      >
    ],
    last_name: "Twain"
  },
  errors: [first_name: {"can't be blank", [validation: :required]}],
  data: #Person<>,
  valid?: false,
  ...
>
This time, we get an error in our changeset, since we no longer provided a first_name.
You can see this difference in the new and new! generated functions as well, which will automatically apply the changes from the changeset to produce a struct for the provided schema.
Person.new(mark_twain)
%Person{
  first_name: "Mark",
  last_name: "Twain",
  parents: [
    %Person.Parent{id: nil, relationship: :father, first_name: "John", last_name: "Clemens"},
    %Person.Parent{id: nil, relationship: :mother, first_name: "Jane", last_name: "Clemens"}
  ]
}
Person.new(mark_twain_bad)
%Person{
  first_name: nil,
  last_name: "Twain",
  parents: [
    %Person.Parent{id: nil, relationship: :father, first_name: "John", last_name: "Clemens"},
    %Person.Parent{id: nil, relationship: :mother, first_name: "Jane", last_name: "Clemens"}
  ]
}
You can see that with new we apply the changes regardless of whether there are any errors present in the changeset.
If we want to raise if an error is present then we can use new! instead.
Person.new!(mark_twain)
%Person{
  first_name: "Mark",
  last_name: "Twain",
  parents: [
    %Person.Parent{id: nil, relationship: :father, first_name: "John", last_name: "Clemens"},
    %Person.Parent{id: nil, relationship: :mother, first_name: "Jane", last_name: "Clemens"}
  ]
}
Person.new!(mark_twain_bad)

  
    
  
  Block Validations


Block validations can be specified using do blocks in field and field!. These are specified as lists of error_condition -> error_message pairs. If the error condition returns true, then the corresponding error_message will be inserted into the changeset when using the generated changeset, new, and new! functions.
Within these validations, you can pass custom bindings, meaning that you can define these validations with respect to variables only available at runtime.
In addition to any bindings you pass, the calues of the fields themselves will be available as a variable with the same name as the field.
You can also refer to local and imported / aliased function within these validations as well.
defmodule Person do
  use Flint.Schema, extensions: []

  def starts_with_capital?(""), do: false

  def starts_with_capital?(<<first::utf8, _rest::binary>>) do
    first in ?A..?Z
  end

  @primary_key false
  embedded_schema do
    field! :first_name, :string do
      !starts_with_capital?(first_name) -> "Must be capitalized!"
      String.length(first_name) >= 10 -> "Name too long!"
    end

    field(:last_name, :string)
  end
end
{:module, Person, <<70, 79, 82, 49, 0, 0, 23, ...>>, :ok}
Person.new!(%{first_name: "mark"})
Person.new!(%{first_name: "Mark"})
%Person{first_name: "Mark", last_name: nil}
All error conditions will be checked, so if multiple error conditions are met then you can be sure that they are reflected in the changeset.
Person.new!(%{first_name: "markmarkmark"})

  
    
  
  Extensions


Flint is designed to be highly extensible and flexible. The main way to extend Flints functionality is through extensions.
Flint currently offers four ways to extend behavior:
	Schema-level attributes
	Field-level additional options
	Default embedded_schema definitions
	Injected Code

Note that extensions are inherited by every child embedded schema defined with embeds_one / embeds_many

  
    
  
  1. Schema-Level Attributes


Extension let you define schema-level attributes which can then be reflected upon later. This is a pattern already used in Ecto such as with the @primary_key attribute, which can then be retrieved with __schema__(:primary_key). Flint simply lets you extend this to any module attribute you want.
However, you can still use this to modify attributes already used by Ecto. Let's take a look at the built-in Embedded extension, which sets attributes to defaults which make more sense when using an embedded_schema rather than a schema, as an example.
defmodule Embedded do
  use Flint.Extension

  attribute :schema_prefix
  attribute :schema_context
  attribute :primary_key, default: false
  attribute :timestamp_opts, default: [type: :naive_datetime]
end

{:module, Embedded, <<70, 79, 82, 49, 0, 0, 47, ...>>,
 [
   Flint.Extension.Dsl.Attributes.Attribute.Options,
   nil,
   %{opts: [], entities: [%Flint.Extension.Field{...}, ...]}
 ]}
When you use the Embedded extension your schema will have these attributes set and can reflect on them.
defmodule ExampleEmbedded do
  use Flint.Schema, extensions: [Embedded]

  embedded_schema do
    field :name, :string
  end
end
{:module, ExampleEmbedded, <<70, 79, 82, 49, 0, 0, 20, ...>>, :ok}
ExampleEmbedded.__schema__(:fields)
[:name]
Notice that there is no :id field defined, which would be set if the @primary_key field were not set to false.

  
    
  
  2. Field-Level Additional Options


This built-in JSON extension is inspired by how marshalling is done in Go, and registers three additional options that can be used to annotate a field. It then uses that option information to define implementations for Jason and Poison Encoder protocols, depending on which you specify.
You can see from that this uses the __schema__(:extra_options) reflection function on the schema, which stores all of the options registered across all extensions.
defmodule JSON do
  use Flint.Extension

  option :name, required: false, validator: &is_binary/1
  option :omitempty, required: false, default: false, validator: &is_boolean/1
  option :ignore, required: false, default: false, validator: &is_boolean/1

  @doc false
  def encode_to_map(module, struct) do
    struct
    |> Ecto.embedded_dump(:json)
    |> Enum.reduce(%{}, fn {key, val}, acc ->
      field_opts = get_field_options(module, key)
      json_key = field_opts[:name] || to_string(key)

      cond do
        field_opts[:ignore] ->
          acc

        field_opts[:omitempty] && is_nil(val) ->
          acc

        true ->
          Map.put(acc, json_key, val)
      end
    end)
  end

  defp get_field_options(module, field) do
    module.__schema__(:extra_options)
    |> Keyword.get(field, [])
    |> Enum.into(%{})
  end

  defmacro __using__(opts) do
    json_module = Keyword.get(opts, :json_module, Jason)
    protocol = Module.concat([json_module, Encoder])

    quote do
      if Code.ensure_loaded?(unquote(json_module)) do
        defimpl unquote(protocol) do
          def encode(value, opts) do
            encoded_map = Flint.Extensions.JSON.encode_to_map(unquote(__CALLER__.module), value)
            unquote(Module.concat([json_module, Encoder, Map])).encode(encoded_map, opts)
          end
        end
      end
    end
  end
end

{:module, JSON, <<70, 79, 82, 49, 0, 0, 60, ...>>, {:__using__, 1}}
Now let's define a schema with specialized serialization using the JSON extension
defmodule ExampleJSON do
  use Flint.Schema, extensions: [JSON]

  @primary_key false
  embedded_schema do
    field! :first_name, :string, name: "First Name"
    field :last_name, :string, name: "Last Name", omitempty: true
    field :nicknames, {:array, :string}, ignore: true
  end
end
{:module, ExampleJSON, <<70, 79, 82, 49, 0, 0, 21, ...>>, :ok}
person =
  ExampleJSON.new!(%{
    first_name: "Charles",
    nicknames: ["Charlie"]
  })
%ExampleJSON{first_name: "Charles", last_name: nil, nicknames: ["Charlie"]}
Jason.encode!(person)
"{\"First Name\":\"Charles\"}"
Here, you can see the effects of each of the 3 options:
	The :first_name field is encoded as "First Name" as specified with the :name option
	The :last_name field is not encoded at all, since it was empty (nil)
	The :nicknames field is not encoded since it is ignored

You can also pass options to extensions. In this case, we can specify a different JSON library to use if we don't want to use Jason, which is the default.
defmodule ExamplePoison do
  use Flint.Schema, extensions: [{JSON, json_module: Poison}]

  @primary_key false
  embedded_schema do
    field! :first_name, :string, name: "First Name"
    field :last_name, :string, name: "Last Name", omitempty: true
    field :nicknames, {:array, :string}, ignore: true
  end
end
{:module, ExamplePoison, <<70, 79, 82, 49, 0, 0, 21, ...>>, :ok}
person =
  ExamplePoison.new!(%{
    first_name: "Charles",
    nicknames: ["Charlie"]
  })
Poison.encode!(person)
"{\"First Name\":\"Charles\"}"

  
    
  
  3. Default embedded_schema Definitions


Extensions also let you define default embedded_schema definitions which will be merged with any schema that uses it.
defmodule Inherited do
  use Flint.Extension

  attribute(:schema_prefix)
  attribute(:schema_context)
  attribute(:primary_key, default: false)
  attribute(:timestamp_opts, default: [type: :naive_datetime])

  embedded_schema do
    field!(:timestamp, :utc_datetime_usec)
    field!(:id)

    embeds_one :child, Child do
      field(:name, :string)
      field(:age, :integer)
    end
  end
end

defmodule Schema do
  use Flint.Schema, extensions: [Inherited]

  embedded_schema do
    field(:type, :string)
  end
end
{:module, Schema, <<70, 79, 82, 49, 0, 0, 23, ...>>, :ok}
Schema.__schema__(:fields)
[:type, :timestamp, :id, :child]
Schema.__schema__(:embeds)
[:child]
defmodule Person do
  use Flint.Schema

  embedded_schema do
    field :first_name
    field :last_name
  end
end
{:module, Person, <<70, 79, 82, 49, 0, 0, 29, ...>>, :ok}
defmodule Event do
  use Flint.Extension

  embedded_schema do
    field!(:timestamp, :utc_datetime_usec)
    field!(:id)
    embeds_one(:person, Person)

    embeds_one :child, Child do
      field(:name, :string)
      field(:age, :integer)
    end
  end
end
{:module, Event, <<70, 79, 82, 49, 0, 0, 47, ...>>, :ok}
defmodule Webhook do
  use Flint.Schema, extensions: [Event, Embedded]

  embedded_schema do
    field :route, :string
    field :name, :string
  end
end
{:module, Webhook, <<70, 79, 82, 49, 0, 0, 25, ...>>, :ok}
Webhook.__schema__(:extensions)
[Event, Flint.Extensions.Embedded]
Webhook.__schema__(:fields)
[:route, :name, :timestamp, :id, :person, :child]
Webhook.__schema__(:embeds)
[:person, :child]

  
    
  
  4. Injected Code


Lastly, extensions can define their own __using__/1 macro that will be called by the schema using the extension.
This is also why the order in which extensions are specified matters, since extensions will be used by the calling schema module in the order that they are specified in the extensions option.
Here's an example of the built-in Accessible extensions, which implements the Access behaviour for the schema.
defmodule Accessible do
  use Flint.Extension

  defmacro __using__(_opts) do
    quote do
      @behaviour Access

      @impl true
      defdelegate fetch(term, key), to: Map
      @impl true
      defdelegate get_and_update(term, key, fun), to: Map
      @impl true
      defdelegate pop(data, key), to: Map
    end
  end
end

{:module, Accessible, <<70, 79, 82, 49, 0, 0, 48, ...>>, {:__using__, 1}}
defmodule AccessibleSchema do
  use Flint.Schema, extensions: [Accessible]

  embedded_schema do
    field :name
  end
end
{:module, AccessibleSchema, <<70, 79, 82, 49, 0, 0, 22, ...>>, :ok}
person = AccessibleSchema.new!(%{name: "Mickey"})
person[:name]
"Mickey"

  
    
  
  Custom Types


Ecto allows you to define custom types by implementing the Ecto.Type or Ecto.ParameterizedType behaviours.
Types are how you define the way in which data is imported (cast / load) and exported (dump) when using your schema. So when accepting external data, types are what determines how that data is put into and taken out of your struct, which is an important behavior to control when working ingesting external data or outputting your data to an external API.
This is common when using Ecto as a means to validate JSON data across programming language barriers.
You might find, however, as you try to write your own types that they can be quite tedious and verbose to implement.
That's where Flint.Type comes in!
Flint.Type is meant to make writing new Ecto types require much less boilerplate, because you can base your
type off of an existing type and only modify the callbacks that have different behavior.
Simply use Flint.Type and pass the :extends option which says which type module to inherit callbacks
from. This will delegate all required callbacks and any implemented optional callbacks and make them
overridable.
It also lets you make a type from an Ecto.ParameterizedType with default parameter values.
You may supply any number of default parameters. This essentially provides a new
init/1 implementation for the type, supplying the default values, while not affecting any of the
other Ecto.ParameterizedType callbacks. You may still override the newly set defaults at the local level.
Just supply all options that you wish to be defaults as extra options when using Flint.Type.
You may override any of the inherited callbacks inherit from the extended module
in the case that you wish to customize the module further.
defmodule Category do
  use Flint.Type, extends: Ecto.Enum, values: [:folder, :file]
end
{:module, Category, <<70, 79, 82, 49, 0, 0, 12, ...>>, :ok}
This will apply default values to Ecto.Enum when you supply a Category type
to an Ecto schema. You may still override the values if you supply the :values
option for the field.
defmodule Downloads do
  use Flint.Schema

  embedded_schema do
    field :type, Category
  end
end
{:module, Downloads, <<70, 79, 82, 49, 0, 0, 29, ...>>, :ok}
Downloads.new!(%{type: :folder})
%Downloads{type: :folder}
Downloads.new!(%{type: :another})
This will create a new NewUID type that behaves exactly like an Ecto.UUID except it dumps
its string length.
import Flint.Type
deftype NewUID, extends: Ecto.UUID, dump: &String.length/1

defmodule TestType do
  use Flint.Schema

  embedded_schema do
    field(:id, NewUID)
  end
end

Ecto.UUID.generate() |> NewUID.dump()
36

  
    
  
  Default Extensions


You can get a list of the default extensions with Flint.default_extensions().
Flint.default_extensions()
[Flint.Extensions.PreTransforms, Flint.Extensions.When, Flint.Extensions.EctoValidations,
 Flint.Extensions.PostTransforms, Flint.Extensions.Accessible, Flint.Extensions.Embedded,
 Flint.Extensions.JSON]
You can optionally provide an :except option to filter which extensions to use.
Flint.default_extensions(except: [When])
[Flint.Extensions.PreTransforms, Flint.Extensions.EctoValidations, Flint.Extensions.PostTransforms,
 Flint.Extensions.Accessible, Flint.Extensions.Embedded, Flint.Extensions.JSON]
When explicitly passing which extensions to use, the default extensions are not automatically included, so you can use Flint.default_extensions to use them in addition to whatever extensions you explicitly use.
defmodule MySchema do
  use Flint.Schema, extensions: Flint.default_extensions(except: [JSON, When]) ++ [JSON]
end
{:module, MySchema, <<70, 79, 82, 49, 0, 0, 19, ...>>,
 {:module, Jason.Encoder.MySchema, <<70, 79, 82, ...>>, {:encode, 2}}}
MySchema.__schema__(:extensions)
[Flint.Extensions.PreTransforms, Flint.Extensions.EctoValidations, Flint.Extensions.PostTransforms,
 Flint.Extensions.Accessible, Flint.Extensions.Embedded, Flint.Extensions.JSON]

  
    
  
  Built-In Extensions


Flint provides a bevy of built-in extensions (those listed in Flint.default_extensions) to provide common conveniences. When building out your own custom Flint extensions, you can refer to the implementation details for any of these extensions for reference.
Let's walk through the different extensions:

  
    
  
  Accessible


An extension to automatically implement the Access behaviour for your struct,
deferring to the Map implementation.
defmodule AccessibleSchema do
  use Flint.Schema, extensions: [Accessible]

  embedded_schema do
    field :name
    embeds_one :embedded, AccessibleEmbed do
      field :type
      field :category
    end
  end
end
{:module, AccessibleSchema, <<70, 79, 82, 49, 0, 0, 24, ...>>, :ok}
a =
  AccessibleSchema.new!(%{
    name: "SampleName",
    embedded: %{type: "SampleType", category: "SampleCategory"}
  })
%AccessibleSchema{
  id: nil,
  name: "SampleName",
  embedded: %AccessibleSchema.AccessibleEmbed{
    id: nil,
    type: "SampleType",
    category: "SampleCategory"
  }
}
a[:name]
"SampleName"
a[:embedded][:category]
"SampleCategory"

  
    
  
  EctoValidations


Shorthand options for common validations found in Ecto.Changeset
Just passthrough the option for the appropriate validation and this extension
will take care of calling the corresponding function from Ecto.Changeset on
your data.
Options
	:greater_than (Ecto.Changeset.validate_number/3)
	:less_than (Ecto.Changeset.validate_number/3)
	:less_than_or_equal_to (Ecto.Changeset.validate_number/3)
	:greater_than_or_equal_to (Ecto.Changeset.validate_number/3)
	:equal_to (Ecto.Changeset.validate_number/3)
	:not_equal_to (Ecto.Changeset.validate_number/3)
	:format (Ecto.Changeset.validate_format/4)
	:subset_of (Ecto.Changeset.validate_subset/4)
	:in (Ecto.Changeset.validate_inclusion/4)
	:not_in (Ecto.Changeset.validate_exclusion/4)
	:is (Ecto.Changeset.validate_length/3)
	:min (Ecto.Changeset.validate_length/3)
	:max (Ecto.Changeset.validate_length/3)
	:count (Ecto.Changeset.validate_length/3)

Aliases
By default, the following aliases are also available for convenience:
config Flint, aliases: [
  lt: :less_than,
  gt: :greater_than,
  le: :less_than_or_equal_to,
  ge: :greater_than_or_equal_to,
  eq: :equal_to,
  ne: :not_equal_to
]
defmodule EctoValidationsSchema do
  use Flint.Schema, extensions: [EctoValidations]

  embedded_schema do
    field! :first_name, :string,  max: 10, min: 5
    field! :last_name, :string, min: 5, max: 10
    field :favorite_colors, {:array, :string}, subset_of: ["red", "blue", "green"]
    field! :age, :integer, greater_than: 0, less_than: max_age
  end
end
{:module, EctoValidationsSchema, <<70, 79, 82, 49, 0, 0, 24, ...>>, :ok}
EctoValidationsSchema.changeset(
  %EctoValidationsSchema{},
  %{first_name: "Bob", last_name: "Smith", favorite_colors: ["red", "blue", "pink"], age: 101},
  max_age: 100
)
#Ecto.Changeset<
  action: nil,
  changes: %{
    first_name: "Bob",
    last_name: "Smith",
    favorite_colors: ["red", "blue", "pink"],
    age: 101
  },
  errors: [
    age: {"must be less than %{number}", [validation: :number, kind: :less_than, number: 100]},
    favorite_colors: {"has an invalid entry", [validation: :subset, enum: ["red", "blue", "green"]]},
    first_name: {"should be at least %{count} character(s)",
     [count: 5, validation: :length, kind: :min, type: :string]}
  ],
  data: #EctoValidationsSchema<>,
  valid?: false,
  ...
>

  
    
  
  Embedded


An extension to house common default configurations for embedded schemas. These configurations are specific for
in-memory schemas.
Attributes
The following attributes and defaults are set by this extension:
	:schema_prefix
	:schema_context
	:primary_key - defaults to false
	:timestamp_opts - defaults to [type: :naive_datetime]

A new schema reflection function is made for each attribute:
__schema__(:schema_context)
...
defmodule WithoutEmbedded do
  use Flint.Schema, extensions: []

  embedded_schema do
    field :name
  end
end
{:module, WithoutEmbedded, <<70, 79, 82, 49, 0, 0, 20, ...>>, :ok}
WithoutEmbedded.__schema__(:fields)
[:id, :name]
WithoutEmbedded.__schema__(:primary_key)
[:id]
defmodule WithEmbedded do
  use Flint.Schema, extensions: [Embedded]

  embedded_schema do
    field :name
  end
end
{:module, WithEmbedded, <<70, 79, 82, 49, 0, 0, 20, ...>>, :ok}
WithEmbedded.__schema__(:fields)
[:name]
WithEmbedded.__schema__(:primary_key)
[]

  
    
  
  JSON


Provides JSON encoding capabilities for Flint schemas with Go-like marshalling options.
This extension enhances Flint schemas with customizable JSON serialization options,
allowing fine-grained control over how fields are represented in JSON output.
Usage
To use this extension, include it in your Flint schema:
  defmodule MySchema do
    use Flint.Schema,
       extensions: [{JSON, json_module: :json}]  # Jason, or Poison
       #extensions: [JSON] # (defaults to Jason if no args passed)

    embedded_schema do
      # Schema fields...
    end
  end
JSON Encoding Options
The following options can be specified for each field in your schema:
	:name - Specifies a custom name for the field in the JSON output.
	:omitempty - When set to true, omits the field from JSON output if its value is nil.
	:ignore - When set to true, always excludes the field from JSON output.

Defining Options
Options are defined directly in your schema using the field macro:
  embedded_schema do
    field :id, :string, name: "ID"
    field :title, :string, name: "Title", omitempty: true
    field :internal_data, :map, ignore: true
  end
defmodule Book do
  use Flint.Schema,
    extensions: [Embedded, JSON]

  embedded_schema do
    field(:id, :string, name: "ISBN")
    field(:title, :string)
    field(:author, :string, omitempty: true)
    field(:price, :decimal, name: "SalePrice")
    field(:internal_notes, :string, ignore: true)
  end
end
{:module, Book, <<70, 79, 82, 49, 0, 0, 23, ...>>, :ok}
book = %{
  id: "978-3-16-148410-0",
  title: "Example Book",
  author: nil,
  price: Decimal.new("29.99"),
  internal_notes: "Not for customer eyes"
}

book |> Book.new!() |> Jason.encode!()
"{\"ISBN\":\"978-3-16-148410-0\",\"SalePrice\":\"29.99\",\"title\":\"Example Book\"}"
You can even specify an alternate JSON module from Jason, such as Poison. In reality, this works with any JSON library that uses a protocol with an encoder implementation to dispatch its JSON encoding. Jason and Poison are the only officially supported ones, both having been tested with the current implementation.
You can specify the JSON library like so:
defmodule PoisonBook do
  use Flint.Schema,
    extensions: [Embedded, {JSON, json_module: Poison}]

  embedded_schema do
    field(:id, :string, name: "ISBN")
    field(:title, :string)
    field(:author, :string, omitempty: true)
    field(:price, :decimal, name: "SalePrice")
    field(:internal_notes, :string, ignore: true)
  end
end
{:module, PoisonBook, <<70, 79, 82, 49, 0, 0, 23, ...>>, :ok}
book |> PoisonBook.new!() |> Poison.encode!()
"{\"title\":\"Example Book\",\"SalePrice\":29.99,\"ISBN\":\"978-3-16-148410-0\"}"

  
    
  
  PreTransforms


The PreTransforms provides a convenient :derive option to express how the field is computed.
By default, this occurs after casting and before validations.
derived fields let you define expressions with support for custom bindings to include any
field declarations that occur before the current field.
:derive will automatically put the result of the input expression into the field value.
By default, this occurs before any other validation, so you can still have access to field
bindings and even the current computed field value (eg. within a :when validation from the
When extension).
You can define a derived field with respect to the field itself, in which case it acts as
transformation. Typically in Ecto, incoming transformations of this support would happen
at the cast step, which means the behavior is determined by the type in which you are casting into.
:derive lets you apply a transformation after casting to change that behavior
without changing the underlying allowed type.
You can also define a derived field with an expression that does not depend on the field,
in which case it is suggested that you use the field macro instead of field! since any input
in that case would be thrashed by the derived value. This means that a field can be completely
determined as a product of other fields!
defmodule Test do
  use Flint.Schema, extensions: [PreTransforms]

  embedded_schema do
    field!(:category, Union, oneof: [Ecto.Enum, :decimal, :integer], values: [a: 1, b: 2, c: 3])
    field!(:rating, :integer)
    field(:score, :integer, derive: rating + category)
  end
end
{:module, Test, <<70, 79, 82, 49, 0, 0, 24, ...>>, :ok}
Test.new!(%{category: 1, rating: 80})
%Test{id: nil, category: 1, rating: 80, score: 81}

  
    
  
  PostTransforms


The PostTransforms extension adds the :map option to Flint schemas.
This works similarly to the PreTransforms extension, but uses the :map option rather than
the :derive option used by PreTransforms, and by default, applies to the field after all validations.
The same caveats apply to the :map expression as all other expressions, with the exception that the
:map function only accepts arity-1 anonymous functions or non-anonymous function expressions
(eg. using variable replacement).
In the following example, :derived is used to normalize incoming strings to downcase to prepare for
the validation, then the output is mapped to the uppercase string using the :map option.
defmodule Character do
  use Flint.Schema

  embedded_schema do
    field! :type, :string, derive: &String.downcase/1, map: String.upcase(type) do
      type not in ~w[elf human] -> "Expected elf or human, got: #{type}"
    end

    field! :age, :integer do
      age < 0 ->
        "Nobody can have a negative age"

      type == "elf" and age > max_elf_age ->
        "Attention! The elf has become a bug! Should be dead already!"

      type == "human" and age > max_human_age ->
        "Expected human to have up to #{max_human_age}, got: #{age}"
    end
  end
end
{:module, Character, <<70, 79, 82, 49, 0, 0, 32, ...>>, :ok}
max_elf_age = 400
max_human_age = 120
Character.new!(%{type: "Elf", age: 10}, binding())

  
    
  
  When


The When extension adds the :when option to Flint schemas.
:when lets you define an arbitrary boolean expression that will be evaluated and pass the validation if it
evaluates to a truthy value. You may pass bindings to this condition and
refer to previously defined fields. :when also lets you refer to the current field in which
the :when condition is defined. Theoretically, you could write many of the other validations using :when, but you will
receive worse error messages with :when than with the dedicated validations.
defmodule WhenTest do
  use Flint.Schema

  embedded_schema do
    field!(:category, Union, oneof: [Ecto.Enum, :decimal, :integer], values: [a: 1, b: 2, c: 3])
    field!(:rating, :integer, when: category == target_category)
    field!(:score, :integer, gt: 1, lt: 100, when: score > rating)
  end
end
{:module, WhenTest, <<70, 79, 82, 49, 0, 0, 31, ...>>, :ok}
WhenTest.new!(%{category: :a, rating: 80, score: 10}, target_category: :a)



  

    
Flint 
    



      
Declarative Ecto embedded_schemas for data validation, coercion, and manipulation.

  
    
  
  Feature Overview


	! Variants of Ecto field, embeds_one, and embeds_many macros to mark a field as required (Required Fields)
	Colocated input transformations let you either transform input fields before validation or derive field values from other fields (Derived Fields / Input Transformations)
	Colocated validations, so you can define common validations alongside field declarations (Validations)
	Colocated output transformations let you transform fields after validation (Mappings / Output Transformations)
	Extensible using the Flint.Extension module. Default extensions include:	Accessible - Adds Access implementation to the target schemas
	JSON - Adds a custom JSON encoding (Jason and Poison supported) implementation to the target schemas
	Embedded - Sets good default module attribute values used by Ecto specifically tailored for in-memory embedded schemas
	And more!


	New Ecto.Schema Reflection Functions	__schema__(:required) - Returns list of fields marked as required (from ! macros)
	And more!


	Convenient generated function (changeset,new,new!,...) (Generated Functions)
	Configurable Application-wide defaults for Ecto.Schema API (Config)
	Conveniently create new Ecto types using the Flint.Type module and its  deftype/2 macro (Flint.Type)


  
    
  
  Installation


def deps do
  [
    {:flint, "~> 0.6"},
    # If you want access to the `Typed` extension to add generated typespecs
    {:typed_ecto_schema, "~> 0.4", runtime: false}
  ]
end

  
    
  
  Motivation


Flint is built on top of Ecto and is meant to provide good defaults for using embedded_schemas for use outside of a database.
It also adds a bevy of convenient features to the existing Ecto API to make writing schemas and validations much quicker.
Of course, since you're using Ecto, you can use this for use as an ORM, but this is emphasizing the use of embedded_schemas as just more expressive and powerful maps while keeping compatibility with Ecto.Changeset, Ecto.Type, and all of the other benefits Ecto has to offer.
In particular, Flint focuses on making it more ergonomic to use embedded_schemas as a superset of Maps, so a Flint.Schema by default implements the Access behaviour and implements the Jason.Encoder protocol.
Flint also was made to leverage the distinction Ecto makes between the embedded representation of the schema and the dumped representation. This means that you can dictate how you want the Elixir-side representation to look, and then provide transformations
for how it should be dumped, which helps when you want the serialized representation to look different.
This is useful if you want to make changes in the server-side code without needing to change the client-side (or vice-versa). Or perhaps you want a mapped representation, where instead of an Ecto.Enum just converting its atom key to a string when dumped, it gets mapped to an integer, etc.

  
    
  
  Basic Usage


If you want to declare a schema with Flint, just use Flint.Schema within your module, and now you have access to Flint's implementation of the
embedded_schema/1 macro.  You can declare an embedded_schema within your module as you otherwise would with Ecto. Within the embedded_schema/1 block, you also have access to Flints implementations of embeds_one,embeds_one!,embeds_many, embeds_many!, field, and field!.
defmodule User do
  use Flint.Schema

  embedded_schema do
    field! :username, :string
    field! :password, :string, redacted: true
    field :nickname, :string
  end
end

  
    
  
  Flint Types


Flint also comes with some types that are automatically aliased when you use Flint.

  
    
  
  Union


Union type for Ecto. Allows the field to be any of the specified types.

  
    
  
  Flint.Type


Flint.Type is meant to make writing new Ecto types require much less boilerplate, because you can base your type off of an existing type and only modify the callbacks that have different behavior.
Simply use Flint.Type and pass the :extends option which says which type module to inherit callbacks
from.  This will delegate all required callbacks and any implemented optional callbacks and make them
overridable.
It also lets you make a type from an Ecto.ParameterizedType with default parameter values.
You may supply any number of default parameters. This essentially provides a new
init/1 implementation for the type, supplying the default values, while not affecting any of the
other Ecto.ParameterizedType callbacks. You may still override the newly set defaults at the local level.
Just supply all options that you wish to be defaults as extra options when using Flint.Type.
You may override any of the inherited callbacks inherity from the extended module
in the case that you wish to customize the module further.

  
    
  
  Examples


defmodule Category do
  use Flint.Type, extends: Ecto.Enum, values: [:folder, :file]
end
This will apply default values to Ecto.Enum when you supply a Category type
to an Ecto schema. You may still override the values if you supply the :values
option for the field.
import Flint.Type
deftype NewUID, extends: Ecto.UUID, dump: &String.length/1
This will create a new NewUID type that behaves exactly like an Ecto.UUID except it dumps
its string length.

  
    
  
  Generated Functions


Flint provides default implementations for the following functions for any schema declaration. Each of these is overridable.
	changeset - Creates a changeset by casting all fields and validating all that were marked as required. If a :default key is provided for a field, then any use of a bang (!) declaration will essentially be ignored since the cast will fall back to the default before any validations are performed.
	new - Creates a new changeset from the empty module struct and applies the changes (regardless of whether the changeset was valid).
	new! - Same as new, except raises if the changeset is not valid.


  
    
  
  Flint Core


The core of Flint is the additional schema macros, which includes the bang (!) variants to mark
a field as required, and the added support of validations through do blocks to fields, as well
as the Flint.Extension API that allows extensions to define additional acceptable field options
and module attributes that can be reflected upon.
All other functionality comes in the form of Flint extensions.
At their core, the new field and field! macros' only additional functionality over the default Ecto macros is to store the allowed Flint options into module attributes which are exposed as new reflection functions.
The bulk of the work done in Flint with validations and transformations of data occurs in the generated changeset function, which leaves it up to the end user whether to use the default implementation, roll their own from scratch using the information exposed through the reflection functions, or do something in between (such as tuning which extensions you use).
When you use Flint.Schema, you declare an overridable changeset function for your schema module that by default just
delegates to the Flint.Changeset.changeset/3 function.
The Flint.Changeset.changeset/3 function operates as the following pipeline:
	Cast all fields (including embeds)
	Validate required fields (Required Fields)


  
    
  
  Required Fields


Flint adds the convenience bang (!) macros (embed_one!,embed_many!, field!) for field declarations within your struct to declare a field as required within its changeset function.
Flint schemas also have a new reflection function in addition to the normal Ecto reflection functions.
	__schema__(:required) -- Returns a list of all fields that were marked as required.


  
    
  
  Field Validations


field do Blocks
In Flint, the field and field! macros now accept an optional do block to define condition/error pairs.
  embedded_schema do
    field! :type, :string do
      type not in ~w[elf human] -> "Expected elf or human, got: #{type}"
    end

    field! :age, :integer do
      age < 0 ->
        "Nobody can have a negative age"

      type == "elf" and age > max_elf_age ->
        "Attention! The elf has become a bug! Should be dead already!"

      type == "human" and age > max_human_age ->
        "Expected human to have up to #{max_human_age}, got: #{age}"
    end
  end
max_elf_age = 400
max_human_age = 120
Character.new!(%{type: "elf", age: 10}, binding())
** (ArgumentError) %Character{type: ["Expected elf or human, got: orc"], age: 10}
    (flint 0.0.1) lib/schema.ex:617: Flint.Schema.new!/3
    (elixir 1.15.7) src/elixir.erl:396: :elixir.eval_external_handler/3
    (stdlib 5.1.1) erl_eval.erl:750: :erl_eval.do_apply/7
    (elixir 1.15.7) src/elixir.erl:375: :elixir.eval_forms/4
    (elixir 1.15.7) lib/module/parallel_checker.ex:112: Module.ParallelChecker.verify/1
    lib/livebook/runtime/evaluator.ex:622: anonymous fn/3 in Livebook.Runtime.Evaluator.eval/4
    (elixir 1.15.7) lib/code.ex:574: Code.with_diagnostics/2
The :do block accepts a list of validation clauses, where is clause is of the form:
failure condition -> Error Message
In the :do block expressions, the same rules apply as mentioned across this documentation. You can pass
bindings to apply to the expression, and field name bindings will automatically be passed to the expression so you can just use the field names as variables.
Additionally, you will have access to all local functions and imported functions that the parent module would have, so you can write expressions as you would in the parent module.
The AST representations of these vaildations are stored in a module attribute and can be retrieved using
__schema__(:blocks)
These validations are checked in the default Flint.Changeset.changeset.
Parameterized Validations
You can even parameterize the options passed to the validations:
defmodule Person do
  use Flint.Schema

  embedded_schema do
    field! :first_name, :string,  max: 10, min: 5
    field! :last_name, :string, min: 5, max: 10
    field :favorite_colors, {:array, :string}, subset_of: ["red", "blue", "green"]
    field! :age, :integer, greater_than: 0, less_than: max_age
  end
end
If you do this, make sure to pass the options as a keyword list into the call to changeset:
Person.changeset(
  %Person{},
  %{first_name: "Bob", last_name: "Smith", favorite_colors: ["red", "blue", "pink"], age: 101},
  [max_age: 100]
)
#Ecto.Changeset<
  action: nil,
  changes: %{
    age: 101,
    first_name: "Bob",
    last_name: "Smith",
    favorite_colors: ["red", "blue", "pink"]
  },
  errors: [
    first_name: {"should be at least %{count} character(s)",
     [count: 5, validation: :length, kind: :min, type: :string]},
    favorite_colors: {"has an invalid entry", [validation: :subset, enum: ["red", "blue", "green"]]},
    age: {"must be less than %{number}", [validation: :number, kind: :less_than, number: 100]}
  ],
  data: #Person<>,
  valid?: false,
  ...
>
Validate With Respect to Other Fields
You might find yourself wishing to validate a field conditionally based on the values of other fields. In Flint, you
can do this with any validation! Since all validations already accept parameterized conditions, they also let you refer
to previously defined fields declared with field or field! macros. Just use a variable of the same name as the field(s) you want to refer to, and they will be bound to their respective variables.

  
    
  
  Extensions


Flint provides an extensible architecture using the Flint.Extension module.
In fact, most of the core features that Flint offers are written as Flint extensions.
The default extensions can be retrieved using Flint.default_extensions()
Flint.default_extensions()

[
  Flint.Extensions.PreTransforms,
  Flint.Extensions.When,
  Flint.Extensions.EctoValidations,
  Flint.Extensions.PostTransforms,
  Flint.Extensions.Accessible,
  Flint.Extensions.Embedded,
  Flint.Extensions.JSON
]
To use extensions, you can specify them when using Flint.Schema in your module. If the :extensions option
is not provided, the default extensions will be used.
defmodule MySchema do
  use Flint.Schema,
    extensions: [Accessible, JSON]

  embedded_schema do
    # Schema fields...
  end
end
Note that you don't have to write the fully-qualified name for modules in the Flint.Extensions module.
You can use Flint.default_extensions() to refer to the default extensions, which you will have to
explicitly add if you pass custom values to the :extensions option when using Flint.Schema, eg.
defmodule MySchema do
  use Flint.Schema,
    extensions: Flint.default_extensions() ++ [MyExtension]

  embedded_schema do
    # Schema fields...
  end
end
You can also create custom extensions by useing Flint.Extension. This allows you to add additional functionality or modify the behavior of Flint schemas according to your specific needs.
For more details on creating and using extensions, refer to the Flint.Extension module documentation.

  
    
  
  Aliases


If you don't like the name of an option, you can provide a compile-time list of aliases to map new option names to existing options
(Validation Options and Transformation Options).
In your config, add an :aliases key with a Keyword value, where each key is the new alias, and the value is an existing option name.
For example, these are default aliases implemented in Flint:
config Flint, aliases: [
    lt: :less_than,
    gt: :greater_than,
    le: :less_than_or_equal_to,
    ge: :greater_than_or_equal_to,
    eq: :equal_to,
    ne: :not_equal_to
  ]
NOTE If you add your own aliases and want to keep these above defaults, you will have to add them manually to your aliases.

  
    
  
  Config


You can configure the default options set by Flint.
	embeds_one: The default arguments when using embeds_one. Defaults to [defaults_to_struct: true, on_replace: :delete]
	embeds_one!: The default arguments when using embeds_one!. Defaults to [on_replace: :delete]
	embeds_many: The default arguments when using embeds_many or embeds_many!. Defaults to [on_replace: :delete]
	embeds_many!: The default arguments when using embeds_many!. Defaults to [on_replace: :delete]
	:aliases: Aliases

You can also configure any aliases you want to use for schema validations.

  
    
  
  Embedded vs Dumped Representations


Flint takes advantage of the distinction Ecto makes between an embedded_schema's embedded and dumped representations.
For example, Flint provides the Flint.Types.Enum type, which are Ecto.Enums where, when given values that are Keyword (rather than just lists of atoms) will have their keys
be the embedded representation and will have the values be the dumped representation.
defmodule Book do
  use Flint.Schema, schema: [
    field(:genre, Flint.Types.Enum, values: [biography: 0, science_fiction: 1, fantasy: 2, mystery: 3])
  ]
end

book = Book.new(%{genre: "biography"})
# %Book{genre: :biography}

Flint.Schema.dump(book)
# %{genre: 0}
In this example, you can see how you can share multiple representations of the same data using this distinction.
You can also implement your own Ecto.Type and further customize this (see Flint.Type):
defmodule ContentType do
  use Ecto.Type
  def type, do: :atom

  def cast("application/json"), do: {:ok, :json}

  def cast(_), do: :error
  def load(_), do: :error

  def dump(:json), do: {:ok, "application/json"}
  def dump(_), do: :error

  def embed_as(_) do
    :dump
  end
end
Here, cast will be called when creating a new Flint schema from a map, and dump will be used
when calling Flint.Schema.dump/1.
defmodule URL do
  use Flint, schema: [
    field(:content_type, ContentType)
  ]
end
url = URL.new(%{content_type: "application/json"})
# %URL{content_type: :json}

Flint.Schema.dump(url)
# %{content_type: "application/json"}

  
    
  
  Examples


You can view the Notebooks folder for some examples in Livebook.
You can also look at Merquery for a real, comprehensive
example of how to use Flint.

Flint.Changeset 
    



      
The base changeset function defined by Flint. Flint.Changeset uses the module attributes
that are collected when using the Flint.Schema macros to perform transformations and validations.

      


      
        Summary


  
    Functions
  


    
      
        changeset(schema, params \\ %{}, bindings \\ [])

      


        Given a Flint (or Ecto) schema and params (can be a map, struct of the given schema, or an existing changeset),
applies all steps of the Flint.Changeset to generate a new changeset.

      Link to this function
    
    changeset(schema, params \\ %{}, bindings \\ [])


      
       
       View Source
     


  


  

Given a Flint (or Ecto) schema and params (can be a map, struct of the given schema, or an existing changeset),
applies all steps of the Flint.Changeset to generate a new changeset.
This function casts all fields (recursively casting all embeds using this same function),
validates required fields (specified using the bang (!) macros exposed by Flint),
outputting the resulting Ecto.Changeset.

  


        

      



  

    
Flint.Schema 
    



      
Flint.Schema provides custom implementations of certain Ecto embedded_schema DSL keywords.
When you use Flint, all of these definitions are imported into the module and override the default
Ecto.Schema implementations. You should not have to directly interact with this module.

        dump(obj)

      


        Dumps the JSON representation of the given schema



    


    
      
        embedded_schema(list)

      


        Wraps Ecto's embedded_schema macro, injecting Flint's custom macro implementation into the module space.



    


    
      
        embeds_many(name, schema, opts \\ [])

      


        Wraps Ecto's embeds_many macro to accept additional options which are consumed by Flint.Changeset.



    


    
      
        embeds_many(name, schema, opts, list)

      


    


    
      
        embeds_many!(name, schema, opts \\ [])

      


        Marks an embeds_many field as required by storing metadata in a module attribute, then calls Flint.Schema.embeds_many.



    


    
      
        embeds_many!(name, schema, opts, list)

      


    


    
      
        embeds_one(name, schema, opts \\ [])

      


        Wraps Ecto's embeds_one macro to accept additional options which are consumed by Flint.Changeset.



    


    
      
        embeds_one(name, schema, opts, list)

      


    


    
      
        embeds_one!(name, schema, opts \\ [])

      


        Marks an embeds_one field as required by storing metadata in a module attribute, then calls Flint.Schema.embeds_one.



    


    
      
        embeds_one!(name, schema, opts, list)

      


    


    
      
        field(name, type \\ :string, opts \\ [])

      


        Wraps Ecto's field macro to accept additional options which are consumed by Flint.Changeset.



    


    
      
        field!(name, type \\ :string, opts \\ [])

      


        Marks a field as required by storing metadata in a module attribute, then calls Flint.Schema.field.



    


    
      
        new(module, params \\ %{}, bindings \\ [])

      


        Creates a new schema struct according to the schema's changeset implementation, immediately applying
the changes from the changeset regardless of errors.



    


    
      
        new!(module, params \\ %{}, bindings \\ [])

      


        Same as new, except will raise if any errors exist in the changeset.

      Link to this function
    
    dump(obj)


      
       
       View Source
     


  


  

Dumps the JSON representation of the given schema

  



  
    
      
      Link to this macro
    
    embedded_schema(list)


      
       
       View Source
     


      (macro)

  


  

Wraps Ecto's embedded_schema macro, injecting Flint's custom macro implementation into the module space.

  



    

  
    
      
      Link to this macro
    
    embeds_many(name, schema, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Wraps Ecto's embeds_many macro to accept additional options which are consumed by Flint.Changeset.
Flint options that are passed to embeds_many are stripped by Flint before passing them to Ecto.Schema.embeds_many and
stored in module attributed for the schema's module.
The following default options are passed to Flint.Schema.embeds_many and can be overriden at the Application level or at the
local call level.

  
    
  
  Default Options


on_replace: :delete

  



  
    
      
      Link to this macro
    
    embeds_many(name, schema, opts, list)

      Link to this macro
    
    embeds_many!(name, schema, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Marks an embeds_many field as required by storing metadata in a module attribute, then calls Flint.Schema.embeds_many.
The metadata tracking is not enforced at the schema / struct level (eg. through enforced keys), but rather
at the schema validation level (through something such as Flint.Changeset.changeset).
The following default options are passed to Flint.Schema.embeds_many! and can be overriden at the Application level or at the
local call level.

  
    
  
  Default Options


on_replace: :delete

  



  
    
      
      Link to this macro
    
    embeds_many!(name, schema, opts, list)


      
       
       View Source
     


      (macro)

  


  


  



    

  
    
      
      Link to this macro
    
    embeds_one(name, schema, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Wraps Ecto's embeds_one macro to accept additional options which are consumed by Flint.Changeset.
Flint options that are passed to embeds_one are stripped by Flint before passing them to Ecto.Schema.embeds_one and
stored in module attributed for the schema's module.
The following default options are passed to Flint.Schema.embeds_one and can be overriden at the Application level or at the
local call level.

  
    
  
  Default Options


defaults_to_struct: true,
on_replace: :delete

  



  
    
      
      Link to this macro
    
    embeds_one(name, schema, opts, list)


      
       
       View Source
     


      (macro)

  


  


  



    

  
    
      
      Link to this macro
    
    embeds_one!(name, schema, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Marks an embeds_one field as required by storing metadata in a module attribute, then calls Flint.Schema.embeds_one.
The metadata tracking is not enforced at the schema / struct level (eg. through enforced keys), but rather
at the schema validation level (through something such as Flint.Changeset.changeset).
The following default options are passed to Flint.Schema.embeds_one! and can be overriden at the Application level or at the
local call level.

  
    
  
  Default Options


on_replace: :delete

  



  
    
      
      Link to this macro
    
    embeds_one!(name, schema, opts, list)


      
       
       View Source
     


      (macro)

  


  


  



    

    

  
    
      
      Link to this macro
    
    field(name, type \\ :string, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Wraps Ecto's field macro to accept additional options which are consumed by Flint.Changeset.
Flint options that are passed to field are stripped by Flint before passing them to Ecto.Schema.field and
stored in module attributed for the schema's module.
If no Flint-specific options or features are used, this acts the same as Ecto.Schema.field.

  



    

    

  
    
      
      Link to this macro
    
    field!(name, type \\ :string, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Marks a field as required by storing metadata in a module attribute, then calls Flint.Schema.field.
The metadata tracking is not enforced at the schema / struct level (eg. through enforced keys), but rather
at the schema validation level (through something such as Flint.Changeset.changeset).

  



    

    

  
    
      
      Link to this function
    
    new(module, params \\ %{}, bindings \\ [])


      
       
       View Source
     


  


  

Creates a new schema struct according to the schema's changeset implementation, immediately applying
the changes from the changeset regardless of errors.
If you want to manually handle error cases, you should use the changeset function itself.

  



    

    

  
    
      
      Link to this function
    
    new!(module, params \\ %{}, bindings \\ [])


      
       
       View Source
     


  


  

Same as new, except will raise if any errors exist in the changeset.

  


        

      



  

    
Flint.Type 
    



      
Flint.Type is meant to make writing new Ecto types require much less boilerplate, because you can base your
type off of an existing type and only modify the callbacks that have different behavior.
Simply use Flint.Type and pass the :extends option which says which type module to inherit callbacks
from.  This will delegate all required callbacks and any implemented optional callbacks and make them
overridable.
It also lets you make a type from an Ecto.ParameterizedType with default parameter values.
You may supply any number of default parameters. This essentially provides a new
init/1 implementation for the type, supplying the default values, while not affecting any of the
other Ecto.ParameterizedType callbacks. You may still override the newly set defaults at the local level.
Just supply all options that you wish to be defaults as extra options when using Flint.Type.
You may override any of the inherited callbacks inherity from the extended module
in the case that you wish to customize the module further.

  
    
  
  Examples


defmodule Category do
  use Flint.Type, extends: Ecto.Enum, values: [:folder, :file]
end
This will apply default values to Ecto.Enum when you supply a Category type
to an Ecto schema. You may still override the values if you supply the :values
option for the field.
import Flint.Type
deftype NewUID, extends: Ecto.UUID, dump: &String.length/1
This will create a new NewUID type that behaves exactly like an Ecto.UUID except it dumps
its string length.

      


      
        Summary


  
    Functions
  


    
      
        deftype(module, opts)

      


        A shorthand for creating a new Flint.Type. This is equivalent to
creating a new module with name module and calling use Flint.Type and passing opts.



    





      


      
        Functions

        


  
    
      
      Link to this macro
    
    deftype(module, opts)


      
       
       View Source
     


      (macro)

  


  

A shorthand for creating a new Flint.Type. This is equivalent to
creating a new module with name module and calling use Flint.Type and passing opts.

  
    
  
  Example


deftype Vehicle, extends: Ecto.Enum, values: [:car, :motorcycle, :truck]

  


        

      



  

    
Flint.Types.Enum 
    




      
        Summary


  
    Functions
  


    
      
        base_type()

      


    


    
      
        cast(arg1, arg2)

      


        Callback implementation for Ecto.ParameterizedType.cast/2.



    


    
      
        dump(arg1, arg2, arg3)

      


        Callback implementation for Ecto.ParameterizedType.dump/3.



    


    
      
        format(arg1)

      


        Callback implementation for Ecto.ParameterizedType.format/1.



    


    
      
        init(opts)

      


        Callback implementation for Ecto.ParameterizedType.init/1.



    


    
      
        load(arg1, arg2, arg3)

      


        Callback implementation for Ecto.ParameterizedType.load/3.



    


    
      
        type(arg1)

      


        Callback implementation for Ecto.ParameterizedType.type/1.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    base_type()


      
       
       View Source
     


  


  


  



  
    
      
      Link to this function
    
    cast(arg1, arg2)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.cast/2.

  



  
    
      
      Link to this function
    
    dump(arg1, arg2, arg3)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.dump/3.

  



  
    
      
      Link to this function
    
    format(arg1)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.format/1.

  



  
    
      
      Link to this function
    
    init(opts)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.init/1.

  



  
    
      
      Link to this function
    
    load(arg1, arg2, arg3)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.load/3.

  



  
    
      
      Link to this function
    
    type(arg1)


      
       
       View Source
     


  


  

Callback implementation for Ecto.ParameterizedType.type/1.

  


        

      



  

    
Flint.Types.Union 
    



      
Union type for Ecto. Allows the field to be any of the specified types.
This is particularly useful if you want to allow one of multiple types and preserve the type of the
input, so long as it adheres to the type (rather than forcing a cast).
For example, suppose you want to allow an input to be a float, or an integer, but you want to
preserve whichever type is passed (keep floats as floats and keep integers as integers), you could
use a Union.

  
    
  
  Options


	:oneof - Allowed types. Can be any valid Ecto type (including custom types). Required.
	:eager - Whether to eagerly cast the field value into the :oneof in the order that
they appear in the list. If false, Flint will attempt to determine if the value maps to a
valid Ecto type before casting and will prefer that type (if it appears in :oneof). Defaults
to false.


  
    
  
  Example


defmodule Character do
  use Flint

  embedded_schema do
    field! :name, :string
    field! :strength, Union, oneof: [:integer, :float]
  end
end
Character.new!(%{name: "Elf", strength: 10})
%Character{name: "Elf", strength: 10}
Character.new!(%{name: "Elf", strength: 10.0})
%Character{name: "Elf", strength: 10.0}

      





  

    
Flint.Extension behaviour
    



      
Flint extensions allow developers to easily hook into Flint metaprogramming lifecycle to add extra data into the embedded
schema reflection functions.
Flint currently offers four ways to extend behavior:
	Schema-level attributes
	Field-level additional options
	Default embedded_schema definitions
	Injected Code

Extension authors define what fields / options / attributes Flint should look for in the module / schema definition and
strip out and store in a schema reflection function, but it is still the resposibility of either the extension author or
the end user to make use of the stored information.

  
    
  
  Schema-Level Attributes


These are simply module attributes that are pre-registered with Flint, and can be given a default value
as well as a validation function. When you use an extension that registers an attribute, then a new __schema__
reflection function is added for each attribute name, with the attribute name as the argument.
Note that the validation occurs at compile time

  
    
  
  Example


Given the following extension:
defmodule Returnable do
  use Flint.Extension

  attribute :returns, validator: fn returns -> is_binary(returns) end
end
And the schema
defmodule Schema do
  use Flint.Schema, extensions: [Returnable]
  @returns "something"
  embedded_schema do
    ...
  end
end
Then you can reflect on this new attribute with:
Schema.__schema__(:returns)

  
    
  
  Field-Level Options


You can also register additional field-level keyword options to be consumed in a downstream function.
These function similarly to the built-in extra options that Flint provides, where the options are
stripped and stored in a module attribute (and subsequently in a __schema__ reflection function)
before passing the valid Ecto.Schema options to Ecto itself.
Note that the validation occurs at compile time

  
    
  
  Example


Given the following extension that enables Go-like JSON marshalling options:
defmodule JSON do
  use Flint.Extension

  option :name, required: false, validator: &is_binary/1
  option :omitempty, required: false, default: false, validator: &is_boolean/1
  option :ignore, required: false, default: false, validator: &is_boolean/1
end
And the following schema:
defmodule Schema do
  use Flint.Schema, extensions: [JSON]
  embedded_schema do
    field :myfield, :string, name: "my_field", omitempty: true
  end
end
Then you can access these specific fields with:
Schema.__schema__(:extra_options)
[
  myfield: [ignore: false, omitempty: true, name: "my_field"],
]

  
    
  
  Default embedded_schema definitions


Sometimes you might want to always have one of more fields defined for a given schema type without having
to write it each time. Much in the same way that, by default, the @prmimary_key attribute will set an :id field
for the schema, you might wish to template out some fields and values.
If you want to define fields that are defined by default in a schema that uses your extension, you can use the
embedded_schema macro within an extension and all field, embeds_one and embeds_many declarations will be
merged into the target schema that uses your extension.
These accept the same arguments as their counterparts in Flint.Schema, and will be
added at the end of the embedded_schema definition.
Duplicate Fields
Note that if you define a template for a field in your extension then anyone who uses your extension will be
unable to override that field name with their own definition.


  
    
  
  Example


Given the following extension:
defmodule Event do
  use Flint.Extension

  embedded_schema do
    field! :timestamp, :utc_datetime_usec
    field! :id, :binary_id
  end
end
and this schema:
defmodule Webhook do
  use Flint.Schema, extensions: [Event, Embedded]

  embedded_schema do
    field :route, :string
  end
end
Then any schema that uses this extension will have these fields by default:
Webhook.__schema__(:fields)
# [:route, :timestamp, :id]

  
    
  
  Injected Code


Lastly, extensions allow you to define custom __using__/1 macros which will be passed through
to the target schema module. This is one of the core functionalities of extensions, and works the
same as you would normally use a module, and helps compartmentalize similar functionality.

  
    
  
  Default Extensions


By default, Flint will enable the following extensions:
	 Flint.Extensions.Block
	 Flint.Extensions.Typed
	 Flint.Extensions.PreTransforms,
	 Flint.Extensions.When,
	 Flint.Extensions.EctoValidations,
	 Flint.Extensions.PostTransforms,
	 Flint.Extensions.Accessible,
	 Flint.Extensions.Embedded,
	 Flint.Extensions.JSON

If you want to pass your own list of extensions for a module, you will need to explicitly pass the defaults
as well if you would like to keep them. You can use the convenience Flint.default_extensions/0 constant
if you want to include all of the defaults.

  
    
  
  Options


	:extensions (list of module that adopts Spark.Dsl.Extension) - A list of DSL extensions to add to the Spark.Dsl

	:otp_app (atom/0) - The otp_app to use for any application configurable options

	:fragments (list of module/0) - Fragments to include in the Spark.Dsl. See the fragments guide for more.



      


      
        Summary


  
    Callbacks
  


    
      
        changeset(t)

      


    


    
      
        changeset(t, keyword)

      


    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    changeset(t)


      
       
       View Source
     


  


  

      

          @callback changeset(Ecto.Changeset.t()) :: Ecto.Changeset.t()


      



  



  
    
      
      Link to this callback
    
    changeset(t, keyword)


      
       
       View Source
     


  


  

      

          @callback changeset(
  Ecto.Changeset.t(),
  keyword()
) :: Ecto.Changeset.t()


      



  


        

      



  

    
Flint.Extensions.Accessible 
    



      
An extension to automatically implement the Access behaviour for your struct,
deferring to the Map implementation.

      





  

    
Flint.Extensions.Block 
    



      
Adds support for do block in field and field! to add validation_condition -> error_message pairs to the field.
Block validations can be specified using do blocks in field and field!. These are specified as lists of error_condition -> error_message pairs. If the error condition returns true, then the corresponding error_message will be inserted into the changeset when using the generated changeset, new, and new! functions.
Within these validations, you can pass custom bindings, meaning that you can define these validations with respect to variables only available at runtime.
In addition to any bindings you pass, the calues of the fields themselves will be available as a variable with the same name as the field.
You can also refer to local and imported / aliased function within these validations as well.

      


      
        Summary


  
    Functions
  


    
      
        changeset(changeset, bindings \\ [])

      


        Uses the quoted expressions from the Flint.Schema.field and Flint.Schema.field!
do blocks to validate the changeset.



    


    
      
        embedded_schema(list)

      


    


    
      
        field(name, type, opts)

      


    


    
      
        field(name, type, opts, list)

      


    


    
      
        field!(name, type, opts)

      


    


    
      
        field!(name, type, opts, list)

      


    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    changeset(changeset, bindings \\ [])


      
       
       View Source
     


  


  

Uses the quoted expressions from the Flint.Schema.field and Flint.Schema.field!
do blocks to validate the changeset.
You can optionally pass bindings to be added to the evaluation context.

  



  
    
      
      Link to this macro
    
    embedded_schema(list)


      
       
       View Source
     


      (macro)

  


  


  



  
    
      
      Link to this macro
    
    field(name, type, opts)


      
       
       View Source
     


      (macro)

  


  


  



  
    
      
      Link to this macro
    
    field(name, type, opts, list)


      
       
       View Source
     


      (macro)

  


  


  



  
    
      
      Link to this macro
    
    field!(name, type, opts)


      
       
       View Source
     


      (macro)

  


  


  



  
    
      
      Link to this macro
    
    field!(name, type, opts, list)


      
       
       View Source
     


      (macro)

  


  


  


        

      



  

    
Flint.Extensions.EctoValidations 
    



      
Shorthand options for common validations found in Ecto.Changeset
Just passthrough the option for the appropriate validation and this extension
will take care of calling the corresponding function from Ecto.Changeset on
your data.

  
    
  
  Options


	:greater_than (Ecto.Changeset.validate_number/3)
	:less_than (Ecto.Changeset.validate_number/3)
	:less_than_or_equal_to (Ecto.Changeset.validate_number/3)
	:greater_than_or_equal_to (Ecto.Changeset.validate_number/3)
	:equal_to (Ecto.Changeset.validate_number/3)
	:not_equal_to (Ecto.Changeset.validate_number/3)
	:format (Ecto.Changeset.validate_format/4)
	:subset_of (Ecto.Changeset.validate_subset/4)
	:in (Ecto.Changeset.validate_inlusion/4)
	:not_in (Ecto.Changeset.validate_exclusion/4)
	:is (Ecto.Changeset.validate_length/3)
	:min (Ecto.Changeset.validate_length/3)
	:max (Ecto.Changeset.validate_length/3)
	:count (Ecto.Changeset.validate_length/3)


  
    
  
  Aliases


By default, the following aliases are also available for convenience:
config Flint, aliases: [
  lt: :less_than,
  gt: :greater_than,
  le: :less_than_or_equal_to,
  ge: :greater_than_or_equal_to,
  eq: :equal_to,
  ne: :not_equal_to
]

  
    
  
  Example


defmodule Person do
  use Flint.Schema

  embedded_schema do
    field! :first_name, :string,  max: 10, min: 5
    field! :last_name, :string, min: 5, max: 10
    field :favorite_colors, {:array, :string}, subset_of: ["red", "blue", "green"]
    field! :age, :integer, greater_than: 0, less_than: max_age
  end
end
Person.changeset(
  %Person{},
  %{first_name: "Bob", last_name: "Smith", favorite_colors: ["red", "blue", "pink"], age: 101},
  [max_age: 100]
)
#Ecto.Changeset<
  action: nil,
  changes: %{
    age: 101,
    first_name: "Bob",
    last_name: "Smith",
    favorite_colors: ["red", "blue", "pink"]
  },
  errors: [
    first_name: {"should be at least %{count} character(s)",
    [count: 5, validation: :length, kind: :min, type: :string]},
    favorite_colors: {"has an invalid entry", [validation: :subset, enum: ["red", "blue", "green"]]},
    age: {"must be less than %{number}", [validation: :number, kind: :less_than, number: 100]}
  ],
  data: #Person<>,
  valid?: false,
  ...
>

      


      
        Summary


  
    Functions
  


    
      
        changeset(changeset, bindings \\ [])

      


        Applies validations to each field according to the options passed in the schema specification.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    changeset(changeset, bindings \\ [])


      
       
       View Source
     


  


  

Applies validations to each field according to the options passed in the schema specification.
See the Field Validations section of the README for more information on validation details.

  


        

      



  

    
Flint.Extensions.Embedded 
    



      
An extension to house common default configurations for embedded schemas. These configurations are specific for
in-memory schemas.

  
    
  
  Attributes


The following attributes and defaults are set by this extension:
	:schema_prefix
	:schema_context
	:primary_key - defaults to false
	:timestamp_opts - defaults to [type: :naive_datetime]

A new schema reflection function is made for each attribute:
__schema__(:schema_context)
...

      





  

    
Flint.Extensions.JSON 
    



      
Provides JSON encoding capabilities for Flint schemas with Go-like marshalling options.
This extension enhances Flint schemas with customizable JSON serialization options,
allowing fine-grained control over how fields are represented in JSON output.

  
    
  
  Usage


To use this extension, include it in your Flint schema:
defmodule MySchema do
  use Flint.Schema,
     extensions: [{JSON, json_module: :json}]  # Jason, or Poison
     #extensions: [JSON] # (defaults to Jason if no args passed)

  embedded_schema do
    # Schema fields...
  end
end

  
    
  
  JSON Encoding Options


The following options can be specified for each field in your schema:
	:name - Specifies a custom name for the field in the JSON output.
	:omitempty - When set to true, omits the field from JSON output if its value is nil.
	:ignore - When set to true, always excludes the field from JSON output.


  
    
  
  Defining Options


Options are defined directly in your schema using the field macro:
embedded_schema do
  field :id, :string, name: "ID"
  field :title, :string, name: "Title", omitempty: true
  field :internal_data, :map, ignore: true
end

  
    
  
  Example


defmodule Book do
  use Flint.Schema,
    extensions: [Embedded, Accessible, JSON]

  embedded_schema do
    field :id, :string, name: "ISBN"
    field :title, :string
    field :author, :string, omitempty: true
    field :price, :decimal, name: "SalePrice"
    field :internal_notes, :string, ignore: true
  end
end

book = %Book{
  id: "978-3-16-148410-0",
  title: "Example Book",
  author: nil,
  price: Decimal.new("29.99"),
  internal_notes: "Not for customer eyes"
}

Jason.encode!(book)
# Results in:
# {
#   "ISBN": "978-3-16-148410-0",
#   "title": "Example Book",
#   "SalePrice": "29.99"
# }
Note how the author field is omitted due to omitempty: true, the internal_notes
field is ignored, and custom names are used for id and price.

  
    
  
  Introspection


You can inspect the JSON options for a schema using the __schema__/1 function:
Book.__schema__(:extra_options)
# Returns:
# [
#   id: [name: "ISBN"],
#   title: [],
#   author: [omitempty: true],
#   price: [name: "SalePrice"],
#   internal_notes: [ignore: true]
# ]
This allows for runtime inspection and manipulation of JSON encoding behavior.

  
    
  
  Implementation Details


This extension implements the Jason.Encoder protocol for your schema, automatically
applying the specified options during JSON encoding. It leverages Flint's schema
introspection capabilities to retrieve and apply the options.
The encoding process:
	Converts the struct to a map
	Applies the :name option to change key names
	Applies the :omitempty option to remove nil values
	Applies the :ignore option to exclude specified fields
	Encodes the resulting map to JSON

This approach provides a flexible and powerful way to control JSON serialization
directly from your schema definition, promoting clean and maintainable code.

      





  

    
Flint.Extensions.PostTransforms 
    



      
The PostTransforms extension adds the :map option to Flint schemas.
This works similarly to the PreTransforms extension, but uses the :map option rather than
the :derive option used by PreTransforms, and by default, applies to the field after all validations.
The same caveats apply to the :map expression as all other expressions, with the exception that the
:map function only accepts arity-1 anonymous functions or non-anonymous function expressions
(eg. using variable replacement).
In the following example, :derived is used to normalize incoming strings to downcase to prepare for
the validation, then the output is mapped to the uppercase string using the :map option.
defmodule Character do
  use Flint.Schema

  embedded_schema do
    field! :type, :string, derive: &String.downcase/1, map: String.upcase(type) do
      type not in ~w[elf human] -> "Expected elf or human, got: #{type}"
    end

    field! :age, :integer do
      age < 0 ->
        "Nobody can have a negative age"

      type == "elf" and age > max_elf_age ->
        "Attention! The elf has become a bug! Should be dead already!"

      type == "human" and age > max_human_age ->
        "Expected human to have up to #{max_human_age}, got: #{age}"
    end
  end
end
max_elf_age = 400
max_human_age = 120
Character.new!(%{type: "Elf", age: 10}, binding())
%Character{type: "ELF", age: 10}

      


      
        Summary


  
    Functions
  


    
      
        changeset(changeset, bindings \\ [])

      


        Applies transformations to each field according to the :map options passed in the schema specification.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    changeset(changeset, bindings \\ [])


      
       
       View Source
     


  


  

Applies transformations to each field according to the :map options passed in the schema specification.
These transformations are applied after validations when used within the default Flint.Changeset.changeset implementation.
Accepts optional bindings which are passed to evaluated code.

  


        

      



  

    
Flint.Extensions.PreTransforms 
    



      
The PreTransforms provides a convenient :derive option to express how the field is computed.
By default, this occurs after casting and before validations.
derived fields let you define expressions with support for custom bindings to include any
field declarations that occur before the current field.
:derive will automatically put the result of the input expression into the field value.
By default, this occurs before any other validation, so you can still have access to field
bindings and even the current computed field value (eg. within a :when validation from the
When extension).
You can define a derived field with respect to the field itself, in which case it acts as
transformation. Typically in Ecto, incoming transformations of this support would happen
at the cast step, which means the behavior is determined by the type in which you are casting into.
:derive lets you apply a transformation after casting to change that behavior
without changing the underlying allowed type.
You can also define a derived field with an expression that does not depend on the field,
in which case it is suggested that you use the field macro instead of field! since any input
in that case would be thrashed by the derived value. This means that a field can be completely
determined as a product of other fields!
defmodule Test do
  use Flint.Schema

  embedded_schema do
    field! :category, Union, oneof: [Ecto.Enum, :decimal, :integer], values: [a: 1, b: 2, c: 3]
    field! :rating, :integer, when: category == target_category
    field :score, :integer, derive: rating + category, gt: 1, lt: 100, when: score > rating
  end
end
Test.new!(%{category: 1, rating: 80}, target_category: 1)

# %Test{category: 1, rating: 80, score: 81}

      


      
        Summary


  
    Functions
  


    
      
        changeset(changeset, bindings \\ [])

      


        Applies transformations to each field according to the :derive options passed in the schema specification.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    changeset(changeset, bindings \\ [])


      
       
       View Source
     


  


  

Applies transformations to each field according to the :derive options passed in the schema specification.
These transformations are applied after casting, but before validations when used within the default Flint.Changeset.changeset implementation.
Accepts optional bindings which are passed to evaluated code.

  


        

      



  

    
Flint.Extensions.Typed 
    



      
 Adds supports for most of the features from the wonderful typed_ecto_schema library.
Rather than using the typed_embedded_schema macro from that library, however, thr Typed extension incorporates the features into the standard
embedded_schema macro from Flint.Schema, meaning even fewer lines of code changed to use typed embedded schemas!
Included with that are the addtional Schema-Level options
you can pass to the embedded_schema macro.
You also have the ability to override field typespecs  as well as providing extra field-level options from
typed_ecto_schema.
Required vs Enforced vs Null
Note that the typespecs allow you to specify :enforce and :null options, which are different from the requirement imposed by field!.

field! marks the field as being required during the
changeset validation, which is equal to passing
the field name to the Eco.Changeset.validate_required/3.
An instance of the schema's struct CAN still be created if a field marked with field! is not present.

enforce: true is equal to including that field in the @enforce_keys module attribute
for the corresponding schema struct.
An instance of the schema's struct CANNOT be created if a field marked with enforce: true is not present, as it
will raise an ArgumentError.

:null indicates whether nil is a valid value for the field. This only impacts the generated typespecs.

Flint.Extensions.When 
    



      
The When extension adds the :when option to Flint schemas.
:when lets you define an arbitrary boolean expression that will be evaluated and pass the validation if it
evaluates to a truthy value. You may pass bindings to this condition and
refer to previously defined fields. :when also lets you refer to the current field in which
the :when condition is defined. Theoretically, you could write many of the other validations using :when, but you will
receive worse error messages with :when than with the dedicated validations.
defmodule Test do
  use Flint.Schema

  embedded_schema do
    field! :category, Union, oneof: [Ecto.Enum, :decimal, :integer], values: [a: 1, b: 2, c: 3]
    field! :rating, :integer, when: category == target_category
    field! :score, :integer, gt: 1, lt: 100, when: score > rating
  end
end
> Test.new!(%{category: :a, rating: 80, score: 10}, target_category: :a)

** (ArgumentError) %Test{category: :a, rating: 80, score: ["Failed `:when` validation"]}
  (flint 0.0.1) lib/schema.ex:406: Flint.Schema.new!/3
  (elixir 1.15.7) src/elixir.erl:396: :elixir.eval_external_handler/3
  (stdlib 5.1.1) erl_eval.erl:750: :erl_eval.do_apply/7
  (elixir 1.15.7) src/elixir.erl:375: :elixir.eval_forms/4
  (elixir 1.15.7) lib/module/parallel_checker.ex:112: Module.ParallelChecker.verify/1
  lib/livebook/runtime/evaluator.ex:622: anonymous fn/3 in Livebook.Runtime.Evaluator.eval/4
  (elixir 1.15.7) lib/code.ex:574: Code.with_diagnostics/2

